

# Exercise 3

Compilation 0368:3133

Due 27/12/2017

## 1 Introduction

We continue our journey of building a compiler for the invented object oriented language RioMare. Remember that the entire specification of RioMare appears inside the relevant folder of the course website. In order to make this document self contained, all the information needed to complete the third exercise is brought here again.

## 2 Programming Assignment

The third exercise implements a semantic analyzer that recursively scans the AST produced by CUP, and checks if it contains any semantic errors. The input for the semantic analyzer is a (single) text file containing a RioMare program, and the output is a (single) text file indicating whether the input program is semantically valid or not. In addition to that, whenever the input program is valid semantically, the semantic analyzer will add meta data to the abstract syntax tree, which is needed for later phases (code generation and optimization). The added meta data content will not be checked in exercises 3, but the best time to design and implement this addition is exercise 3.

## 3 The RioMare Semantics

This section describes the semantics of RioMare, and provides a multitude of legal and illegal example programs.

### 3.1 Types

The RioMare programming language defines two native types: integers and strings. In addition, it is possible to define a class by specifying its data members and methods. Also, given an existing type  $T$ , one can define an array of  $T$ 's. Note, that defining classes and arrays is only possible in the uppermost (global) scope. The exact details follow.

### 3.1.1 Classes

Classes contain data members and methods, and can only be defined in the uppermost (global) scope. They can refer to/extend only previously defined classes, to ensure that the class hierarchy has a tree structure. Following the same concept, a method M1 can *not* refer to a method M2, whenever M2 is defined after M1 in the class. In contrast to all that, a method M *can* refer to a data member d, even if d is defined *after* M in the class. Table 1 summarizes these facts.

1	<pre>CLASS Son EXTENDS Father {     int bar; } CLASS Father {     void foo() { PrintInt(8); } }</pre>	ERROR
2	<pre>CLASS Edge {     Vertex u;     Vertex v; } CLASS Vertex {     int weight; }</pre>	ERROR
3	<pre>CLASS UseBeforeDef {     void foo() { bar(8); }     void bar(int i) { PrintInt(i); } }</pre>	ERROR
4	<pre>CLASS UseBeforeDef {     void foo() { PrintInt(i); }     int i; }</pre>	OK

Table 1: Referring to classes, methods and data members

**Methods overloading** is *illegal* in RioMare, with the obvious exception of overriding a method in a derived class. Similarly, it is illegal to define a variable with the same name of a previously defined variable (shadowing), or a previously defined method. Table 2 summarizes these facts.

1	<pre> CLASS Father {     int foo() { return 8; } } CLASS Son EXTENDS Father {     void foo() { PrintInt(8); } } </pre>	ERROR
2	<pre> CLASS Father {     int foo(int i) { return 8; } } CLASS Son EXTENDS Father {     int foo(int j) { return j; } } </pre>	OK
3	<pre> CLASS IllegalSameName {     void foo() { PrintInt(8); }     void foo(int i) { PrintInt(i); } } </pre>	ERROR
4	<pre> CLASS Father {     int foo; } CLASS Son EXTENDS Father {     string foo; } </pre>	ERROR

Table 2: Method overloading and variable shadowing are both illegal in RioMare.

**Inheritance** if class `Son` is derived from class `Father`, then any place in the program that semantically allows an expression of type `Father`, should semantically allow an expression of type `Son`. For example,

<pre> CLASS Father { int i; } CLASS Son EXTENDS Father { int j; } void foo(Father f) { PrintInt(f.i); } void main(){ foo(NEW Son); } </pre>	OK
---	----

Table 3: new Son is a semantically valid input for foo.

**Null expressions** any place in the program that semantically allows an expression of type class, should semantically allow `NULL` instead. For instance,

<pre> CLASS Father { int i; } void foo(Father f){ PrintInt(f.i); } void main(){ foo(NULL); } </pre>	OK
---	----

Table 4: `NULL` sent instead of a (Father) class is semantically allowed.

### 3.1.2 Arrays

Arrays can only be defined in the uppermost (global) scope. They are defined with respect to some previously defined type, as in the following example:

```
ARRAY IntArray = int[]
```

Defining an integer matrix, for example, is possible as follows:

```
ARRAY IntArray = int[] ARRAY IntMat = IntArray[]
```

In addition, any place in the program that semantically allows an expression of type array, should semantically allow `NULL` instead. For instance,

<pre> ARRAY IntArray = int[] void F(IntArray A){ PrintInt(A[8]); } void main(){ F(NULL); } </pre>	OK
---	----

Table 5: `NULL` sent instead of an integer array is semantically allowed.

**Note** that allocating arrays with the new operator must be done with an *integral size*. Similarly, accessing an array entry is semantically valid only when the *subscript expression has an integer type*. Note further that if two arrays of type `T` are defined, they are *not* interchangeable:

<pre> ARRAY gradesArray = int[] ARRAY IDsArray    = int[] void F(IDsArray ids){ PrintInt(ids[6]); } void main() {     IDsArray ids      := NEW int[8];     gradesArray grades := NEW int[8];     F(grades); } </pre>	ERROR
--	-------

Table 6: Non interchangeable array types.

### 3.2 Assignments

Assigning an expression to a variable is clearly legal whenever the two have the same type. In addition, following the concept in 3.1.1, if class **Son** is derived from class **Father**, then a variable of type **Father** can be assigned an expression of type **Son**. Furthermore, following the concept in 3.1.1 and 3.1.2, assigning **NULL** to array and class variables is legal. In contrast to that, assigning **NULL** to string variables is *illegal*. To avoid an overly complex semantics, we will enforce a strict policy of initializing data members inside classes: a declared data member inside a class can be initialized *only with a constant value* (that matches its type). Specifically, only constant integers, strings and **NULL** can be used, and even a simple expression like  $5 + 6$  is forbidden. Table 7 summarizes these facts.

1	CLASS Father { int i; } Father f := NULL;	OK
2	CLASS Father { int i; } CLASS Son EXTENDS Father { int j; } Father f := NEW Son;	OK
3	CLASS Father { int i; } CLASS Son EXTENDS Father { int j := 8; }	OK
4	CLASS Father { int i := 9; } CLASS Son EXTENDS Father { int j := i; }	ERROR
5	CLASS Father { int foo() { return 90; } } CLASS Son EXTENDS Father { int j := foo(); }	ERROR
6	CLASS IntList { int head := -1; IntList tail := NEW IntList; }	ERROR
7	CLASS IntList { void Init() { tail := NEW List; } int head; IntList tail; }	OK
8	ARRAY gradesArray = int[] ARRAY IDsArray = int[] IDsArray := NEW int[8]; gradesArray := NEW int[8]; gradesArray := IDsArray;	ERROR
9	string s := NULL;	ERROR

Table 7: Assignments.

### 3.3 If and While Statements

The type of the condition inside if and while statements is the primitive type int.

### 3.4 Equality Testing

Testing equality between two expressions is legal whenever the two have the same type. In addition, following the same reason in 3.2, if class **Son** is derived from class **Father**, then an expression of type **Father** can be tested for equality with an expression of type **Son**. Furthermore, any class variable or array variable can be tested for equality with **NULL**. But, in contrast, it is *illegal* to compare a string variable to **NULL**. The resulting type of a semantically valid comparison is the primitive type **int**. Table 8 summarizes these facts.

1	<pre> CLASS Father { int i; int j; } int Check(Father f) {     if (f = NULL)     {         return 800;     }     return 774; } </pre>	OK
2	<pre> int Check(string s) {     return s = "LosPollosHermanos"; } </pre>	OK
3	<pre> ARRAY gradesArray = int[] ARRAY IDsArray    = int[] IDsArray          := NEW int[8]; gradesArray := NEW int[8]; int i := gradesArray = IDsArray; </pre>	ERROR
4	<pre> string s1; string s2 := "HankSchrader"; int i := s1 = s2; </pre>	OK

Table 8: Equality testing.

### 3.5 Binary Operations

Most binary operations ( $-$ ,  $*$ ,  $/$ ,  $<$ ,  $>$ ) are performed only between integers. The single exception to that is the  $+$  binary operation, that can be performed between integers *and strings*. Note that equality testing ( $=$ ) is treated differently, as discussed in 3.4. The resulting type of a semantically valid binary operation is the primitive type `int`. Table 9 summarizes these facts.

1	<pre>CLASS Father {     int foo() { return 8/0; } }</pre>	OK
2	<pre>CLASS Father { string s1; string s2; } void foo(Father f) {     f.s1 := f.s1 + f.s2; }</pre>	OK
3	<pre>CLASS Father { string s1; string s2; } void foo(Father f) {     int i := f.s1 &lt; f.s2; }</pre>	ERROR
4	<pre>CLASS Father { int j; int k; } int foo(Father f) {     int i := 620;     return i &lt; f.j; }</pre>	OK

Table 9: Binary Operations.



### 3.6 Scope Rules

**RioMare** defines four kinds of scopes: block scopes of if and while statements, function scopes, class scopes and the outermost global scope. When an identifier is being used at some point in the program, its declaration is searched for in all of its enclosing scopes. The search starts from the innermost scope, and ends at the outermost (global) scope.

**Note** that array type declarations and class type declarations can only be defined in the outermost (global) scope, so this subsection is really about functions and variables.

**Functions** can be defined only in the class scopes, and the global scope. When a function is being called inside a class scope, the declaration of a function with that name is searched first in its class scope. If no such function is found, the search moves to the global scope, and if the declaration is missing there too, a semantic error is issued. Following the same reason, when a function is being called inside the global scope, only the global scope is searched for its declaration.

**Resolving** a variable identifier follows the same principal, with the slight difference that variables can be declared in all four kinds of scopes. Table 10 summarizes these facts.

## 4 Input

The input for this exercise is a single text file, the input RioMare program.

## 5 Output

The output is a *single* text file that contains a *single* word. Either OK when the input program is correct semantically, or otherwise ERROR(*location*), where *location* is the line number of the *first* error that was encountered.

## 6 Submission Guidelines

The skeleton code for this exercise resides (as usual) in subdirectory EX3 of the course repository. COMPILATION/EX3 should contain a makefile building your source files to a runnable jar file called COMPILER (note the lack of the .jar suffix). Feel free to use the makefile supplied in the course repository, or write a new one if you want to. Before you submit, make sure that your exercise compiles and runs on the school server: *nova.cs.tau.ac.il*. This is the formal running environment of the course.

**Execution parameters** compiler receives 2 input file names:

InputRioMareProgram.txt  
OutputStatus.txt

1	<pre> int salary := 7800; void foo() {     string salary := "6950"; } </pre>	OK
2	<pre> int salary := 7800; void foo(string salary) {     PrintString(salary) } </pre>	OK
3	<pre> void foo(string salary) {     int salary := 7800;     PrintString(salary) } </pre>	ERROR
4	<pre> string myvar := "80"; CLASS Father {     Father myvar := NULL;     void foo()     {         int myvar := 100;         PrintInt(i);     } } </pre>	OK
4	<pre> int foo(string s) { return 800;} CLASS Father {     string foo(string s)     {         return s;     }     void Print()     {         PrintString(foo());     } } </pre>	OK

Table 10: Scope Rules.

Program ::= dec<sup>+</sup>  
 dec ::= funcDec | varDec | classDec | arrayDec  
 varDec ::= ID ID [ ASSIGN exp ] ';' ;  
 funcDec ::= ID ID '(' [ ID ID [ ',' ID ID ]\* ] ')' '{' stmt [ stmt ]\* '}'  
 classDec ::= CLASS ID [ EXTENDS ID ] '{' cField [ cField ]\* '}'  
 arrayDec ::= ARRAY ID = ID '[' exp ']'  
 exp ::= var  
 exp ::= '(' exp ')'  
 exp ::= exp BINOP exp  
 exp ::= [ var '.' ] ID '(' [ exp [ ',' exp ]\* ] ')'  
 exp ::= '[' '-' ] INT | NIL | STRING | NEW ID | NEW ID '[' exp ']'  
 var ::= ID  
 var ::= var '.' ID  
 var ::= var '[' exp ']'  
 stmt ::= varDec  
 stmt ::= var ASSIGN exp ';' ;  
 stmt ::= RETURN [ exp ] ';' ;  
 stmt ::= IF '(' exp ')' '{' stmt [ stmt ]\* '}'  
 stmt ::= WHILE '(' exp ')' '{' stmt [ stmt ]\* '}'  
 stmt ::= [ var '.' ] ID '(' [ exp [ ',' exp ]\* ] ')';  
 cField ::= varDec | funcDec  
 BINOP ::= + | - | \* | / | < | > | =  
 INT ::= [1 - 9][0 - 9]\* | 0

Table 11: Context free grammar for the RioMare programming language.

Precedence	Operator	Description	Associativity
1	::=	assign	
2	=	equals	left
3	<, >		left
4	+, -		left
5	*, /		left
6	[	array indexing	
7	(	function call	
8	.	field access	left

Table 12: Binary operators of RioMare along with their associativity and precedence. 1 stands for the lowest precedence, and 9 for the highest.